

EE332 Lab2: Divide-Conquer Implementation of 16-4 Priority Encoder

1st Qiu Kunyuan *EEE, Southern University of Science and Technology*
Shenzhen, PRC
11913019@mail.sustech.edu.cn

Abstract

This report focuses on the difference between cascaded structure and tree structure in the Verilog implementation of 16-4 priority encoder. The timing optimization of combinational logic circuits with a large number of inputs is carried out by the divide and conquer method, and the pipelining method is also employed. Therefore, the routing results are obtained with satisfactory delays and no race-hazard jitters.

Code for this experiment can be found at <https://github.com/KagaJiankui/EE332-2024S/tree/master/lab3/lab3.srcs>.

Index Terms

FPGA, programmable logic, priority coder, divide-conquer method, pipelining, -hazard condition, jitter, delay

I. INTRODUCTION

There are typically two main implementations of priority encoders in Verilog designs: cascaded architecture and tree architecture.

- **Cascaded architecture:** A cascaded architecture is a serial structure that groups input signals and encodes them within each group, and then these results are cascaded. Such a structure enjoys the simplicity of implementation, but suffers from the disadvantage that the delay increases with the increase in the number of input signals.
- **Tree architecture:** A tree architecture is a parallel structure that encodes all the input signals simultaneously. Benefit of this structure is its speed as all the operations are performed in parallel. However, the demerit of this structure is the requirement of more hardware resources.

A. Cascaded Architecture

The code of cascaded priority encoder is if-else statements organized along with the order of MSB to LSB. Thus, the cascaded priority encoder can be easily modeled by the following Verilog code with parameter digit:

```

1 module encoder
2   #(
3     parameter digit=8
4     parameter digit_output=3
5   )
6   (
7     input [digit-1:0] x
8     input en,
9     output reg [digit_output-1:0] y
10  );
11  integer i;
12  always @(x or en) begin
13    if (en) begin
14      y = 0;
15      for( i = 0; i <= digit - 1; i = i+1)
16        if(x[i] == 1) y = i[digit-1:0];
17    end
18    else y = 0;
19  end
20 endmodule

```

For 8-3 encoder, the module is instantiated with parameter .digit(8), .digit_out(3)(note that the width of input and output bus is designated separately since Verilog, as an HDL, lacks exponential operation even in parameterization) that generates the following if-else statements,

```

1 module encoder (
2     input [7:0] x,
3     input en,
4     output reg [2:0] y,
5     output reg v
6 );
7 always @ (x or en)
8     if (x>0 & en==1) begin
9         if (x[7]) y = 3'b111;
10        else if (x[6]) y = 3'b110;
11        else if (x[5]) y = 3'b101;
12        else if (x[4]) y = 3'b100;
13        else if (x[3]) y = 3'b011;
14        else if (x[2]) y = 3'b010;
15        else if (x[1]) y = 3'b001;
16        else if (x[0]) y = 3'b000;
17        else y = 3'b000;
18    end
19    else begin
20        y = 0;
21        v = 0;
22    end
23 endmodule

```

and the schematic (1) of the elaboration result is basically directly translated from the HDL description, which are cascaded MUXs.




Fig. 1: Synthesis result of parameterized 8-3 priority encoder

For 4-2 encoder, the module is instantiated with parameter .digit(4), .digit_out(2),

```

1 `timescale 1us / 10ns
2 module encoder4x2_cas (
3     input wire [3:0] in,
4     output reg [1:0] out,
5     output reg v
6 );
7
8 always @(in) begin
9     v = in[3] | in[2] | in[1] | in[0];
10    if (in[3]) begin
11        out = 2'b11;
12    end else if (in[2]) begin
13        out = 2'b10;
14    end else if (in[1]) begin
15        out = 2'b01;
16    end else if (in[0]) begin
17        out = 2'b00;
18    end else begin
19        out = 2'b00;
20    end
21 end
22
23 endmodule

```

which is elaborated into similar schematic (2):




Fig. 2: Synthesis result of parameterized 4-2 priority encoder

B. Tree Architecture

The encoder organized in tree architecture is a reverse binary tree or quadruple tree, as depicted in left of Fig.(3). The leaf nodes at the bottom level are priority encoders with low input width, while the nodes at higher level are usually MUXs and gates that combines the output from bottom level leaf nodes with their priority.




Fig. 3: Architecture Depiction

The internal implementation of the combinational logic block is shown in right of Fig.(3). an encoder layer and the adjacent upper combinational logic node, the combinational logic node consists of a set of multiplexers corresponding to the outputs of each bit from each encoder, as well as an encoder generating a line-select signal, with an input width equal to the number of encoders.

In the designs with building block encoder similar to 74HC148, an encoder IC with *validity* output, the output is therefore composed by the output of the building blocks:

$$\mathbf{valid}[i] = \mathbf{OR}(\mathbf{input}[i]) \quad (1)$$

$$\mathbf{Q} = i * W_{bb} + \mathbf{output}[\mathbf{valid}] \quad (2)$$

where the bolded signal name represents bus, the block bracket represents indexing from bus, the variable i denotes the valid encoder (the location of the highest input bit), and the variable W_{bb} denotes the input width of the building block encoder.

For a priority encoder with input bitwidth W_{input} , output bitwidth W_{output} , it is obvious that the output bitwidth is the bottom 2 logarithm of the input bitwidth. To employ the tree architecture, the input is splitted the input bitwidth into $S = W_{input}/W_{bb}$ slices.

$$W_{output} = \log_2 W_{input} \quad (3)$$

$$\Rightarrow W_{output} - \log_2 W_{bb} = \log_2 \left(\frac{W_{input}}{W_{bb}} \right) \quad (4)$$

$$\Leftrightarrow \begin{cases} \log_2 W_{bb} = W_{output} - \log_2 S, \\ \log_2 S = W_{slice} \end{cases} \quad (5)$$

Therefore, by converting the multiplication in Eq.(2) into bitwise operations, with respect to the power-2 integrity of S and W, the derivation is clear, that the lower W_{bb} bits of the output bit is the encoded output of each slice, and the higher W_{slice} bits is the serial number of the building block encoder with valid state.

For a 16-4 priority encoder in tree architecture, with the building block selected as 4-2 priority encoder with validity indication port, all the parameters can be evaluated from Eq.(5)

$$W_{output} = \log_2 16 = 4, S = 16/4 = 4, W_{slice} = 2, \log_2 W_{bb} = W_{output} - W_{slice} = 2 \quad (6)$$

Then the verilog modeling is performed intuitively:

```

1  `timescale 1us / 10ns
2  module encoder16x4 (
3      input wire clk,
4      input [15:0] in,
5      output reg [3:0] out,
6      output reg v
7  );
8      wire [1:0] enc1, enc2, enc3, enc4;
9      wire v1, v2, v3, v4;
10
11     encoder4x2 e1 (
12         .in (in[3:0]),
13         .out(enc1),
14         .v (v1)
15     );
16     encoder4x2 e2 (
17         .in (in[7:4]),
18         .out(enc2),
19         .v (v2)
20     );
21     encoder4x2 e3 (
22         .in (in[11:8]),
23         .out(enc3),
24         .v (v3)
25     );
26     encoder4x2 e4 (
27         .in (in[15:12]),
28         .out(enc4),
29         .v (v4)
30     );
31
32     reg [3:0] result0, result1, v_sel;
33
34     always @(posedge clk) begin
35         v_sel <= {v4, v3, v2, v1};
36         result1 <= {enc1[1], enc2[1], enc3[1], enc4[1]};
37         result0 <= {enc1[0], enc2[0], enc3[0], enc4[0]};
38     end
39
40     wire [1:0] sel_output;
41     wire v_from_input;
42
43     encoder4x2 e_select (
44         .in (v_sel),
45         .out(sel_output),
46         .v (v_from_input)
47     );
48
49     always @(posedge clk) begin
50         out[3:2] <= sel_output;
51         v <= v_from_input;
52         case (sel_output)
53             2'b11: out[1:0] <= {result1[0], result0[0]};

```

```

54 2'b10: out[1:0] <= {result1[1], result0[1]};
55 2'b01: out[1:0] <= {result1[2], result0[2]};
56 2'b00: out[1:0] <= {result1[3], result0[3]};
57 default: out[1:0] <= 2'b00;
58 endcase
59 end
60 endmodule

```

which results in the Fig.(4) after elaboration, and Fig.(5) after synthesis:




Fig. 4: Elaborated schematic of 16-4 encoder




Fig. 5: Synthesized schematic of 16-4 encoder

II. TIME COMPLEXITY DISCUSSION AND OPTIMIZATION

The tree-structured prioritized encoder applies the idea of division and conquering method, which breaks down the big problem into small problems, and then combines the solutions of the small problems into the solution to the big problem. Each layer of the encoder works in parallel, effectively reducing the output delay.

A. Elaboration Stage Analysis

With the derivation in the section above, it is possible to portrait a model (6) of the internal data flow in the tree architecture, for the evaluation of the total time complexity (device gate delay):




Fig. 6: Data flow and time complexity of the internal

It's obvious that the time complexity of the cascaded architecture priority encoder is $\mathcal{O}(W_{in})$. If the division is only made once, which is equivalent to building block encoder and line selection generator are confined to cascaded architecture, the total delay is therefore obtained:

$$D = \mathcal{O}(W_{bb}) + \mathcal{O}(S) = \mathcal{O}\left(\frac{W_{input}}{S} + S\right) \quad (7)$$

$$\Rightarrow S = \arg \min_S \left(\frac{W_{input}}{S} + S \right) \quad (8)$$

$$= W_{input}^{\frac{1}{2}}, \quad (9)$$

$$D_{min} = 2S = 2W_{input}^{\frac{1}{2}} \quad (10)$$

For divide-conquer implementation with large width input bus, the cascaded architecture delay S itself is unsatisfying, suggesting that the N -branch tree architecture (7) should be applied.




Fig. 7: N-branch tree architecture

With the illustration of Fig.7 (b)) and respect to the power-2 integrity of W, S, N , take the notion at the output side $W = 2^w, S = 2^s, N = 2^n$ and then obtain the following time complexity objective:

$$TC(s, n) = \mathcal{O}\left(\frac{W}{S} + L \cdot N\right) \quad (11)$$

$$= \mathcal{O}\left(2^{w-s} + 2^n \frac{s}{n}\right) \quad (12)$$

Since there is no implied constraints and both n and s are positive integers, the extremum can be directly derived by Hessian matrix:

$$\arg \min_{(s,n)} TC(s, n) = \arg_{(s,n)} \left(\nabla \left(\frac{W}{S} + L \cdot N \right) = 0 \right) \quad (13)$$

$$\Rightarrow (s, n) = (w - \log_2 e, \log_2 e) \quad (14)$$

and the extremum point satisfies the multiple variable minima condition

$$\det H|_{(s,n)} = \nabla(TC(i, j)) \nabla^T(TC(i, j))|_{(s,n)} > 0 \quad (15)$$

Applying the integrity constraints to the solution in real domain $(s, n) = (w - \log_2 e, \log_2 e)$, further discussion reveals

- For variable n , rounding to $n = 1$ or $n = 2$ results in the same value $2^n/n = 2$;
- For variable s , take $2^n/n = 2$ above results in $s = w - \lfloor \log_2(\log_2(e)) + 1 \rfloor = w - 1$

Therefore in conclusion,

$$\arg \min_{(s,n)} TC(s, n), \{n, s, w\} \subseteq \mathbb{Z}^+ \Rightarrow \begin{cases} n \in \{1, 2\} \\ s = w - 1 \end{cases} \quad (16)$$

B. After Synthesis Stage Analysis

Vivado synthesizer optimizes the circuit into LUTs, analog switches and programmable ROMs after the synthesis. For low complexity modules like 4-2 priority encoder (8), either the cascaded implementation or the simplified combinational logic (using Carnot Diagram) are synthesized into a couple of LUTs with unit gate delay. For LUT implementation, the submodule is synthesized into an ROM, also with unit gate delay.




Fig. 8: Alternative implementation of 4-2 encoder submodule

Thus in the time complexity object Eq.(11), the unit submodule delay suggests that

$$\frac{W}{S} = 1, N = 1 \Rightarrow TC(s, n) = \mathcal{O}(L) = \mathcal{O}\left(\frac{s}{n}\right) \quad (17)$$

To minify the possibility of outbounded length in critical paths produced by synthesizer, the input widths W_{bb} are selected to 4, and the number of layers is $L = 1$. Therefore, parameters selection (6) is still appropriate for the 16-4 priority encoder design.




Fig. 9: Timing simulation after synthesis of 4-2 priority encoder




Fig. 10: Timing simulation after synthesis of 16-4 priority encoder, using combinational logic submodule

III. SIMULATION RESULTS

With the elaboration and discussion above, the modeling and simulation are pretty straightforward. Figures below shows the simulation result of 4-2 encoder submodule and the final 16-4 encoder submodule:

In Fig.(9), signal result1 is the output of cascaded architecture submodule, result2 is the output of LUT submodule, and signal result3 is the output of combinational logic (Carnot simplification on design time) submodule.

In Fig.(10), the consistency of the 16-4 priority encoder (with 2 layer pipelining) is verified. Note the time required for the initialization of registers at the beginning.

In Fig.(11), the consistency of the 16-4 priority encoder (with 2 layer pipelining) is verified, the identical critical path between the combinational logic design and the ROM LUT design after synthesis results in identical waveform. Note the time required for the initialization of registers at the beginning.

IV. APPENDIX. CODE

A. Testbench for 16-4 priority encoder




Fig. 11: Timing simulation after synthesis of 16-4 priority encoder, using LUT submodule


```

1  `timescale 1us / 10ns
2  /*
3   Company:
4   Engineer:
5
6   Create Date: 03/19/2024 04:49:04 PM
7   Design Name:
8   Module Name: tb_encoder_16to4
9   Project Name:
10  Target Devices:
11  Tool Versions:
12  Description:
13
14  Dependencies:
15
16  Revision:
17  Revision 0.01 - File Created
18  Additional Comments:
19  */
20
21
22
23 module tb_encoder_16x4_lut;
24     reg [15:0] data;
25     reg [15:0] mask;
26     reg [15:0] test_rand;
27     reg [15:0] incr;
28     wire [3:0] result_cas, result_lut;
29     wire v_cas, v_lut;
30     reg clock;
31     parameter reg [32:0] powerupDelay = 30;
32
33     encoder16x4_cas encoder (
34         .clk(clock),
35         .in (data),
36         .out(result_cas),
37         .v  (v_cas)
38     );
39
40     encoder16x4_lut encoder_lut (
41         .clk(clock),
42         .in (data),
43         .out(result_lut),
44         .v  (v_lut)
45     );
46
47     initial begin
48         clock = 1'b0;
49         data = 16'h0000;
50         incr = 16'h0001;
51         mask = 16'h0000;
52         test_rand = {$urandom} & 16'hFFFF;
53         #powerupDelay; // register initialization on powerUp
54         forever begin
55             #5;
56             clock = ~clock;
57         end
58     end
59
60     always @(posedge clock) begin
61         #2;
62         if (mask == 16'hffff) begin
63             incr = 16'h0001;
64             mask = 16'h0000;
65         end
66         incr = incr << 1;
67         test_rand = {$urandom} & 16'hFFFF;
68         mask = incr - 16'h0001;
69         data = test_rand & mask;
70     end
71
72 endmodule

```

B. Testbench for 4-2 priority encoder

```

1  `timescale 1us / 10ns
2
3
4  /* verilator lint_off UNOPTFLAT */
5
6  module tb_encoder_4to2;
7  reg [3:0] data;
8  wire [1:0] result1;
9  wire [1:0] result2;
10 wire [1:0] result3;
11 wire v;
12
13 encoder4x2_cas enc_cas(
14     .in(data),
15     .out(result1),
16     .v(v)
17 );
18
19 encoder4x2_lut enc_lut(
20     .in(data),
21     .out(result2),
22     .v(v)
23 );
24
25 encoder4x2 enc_tree(
26     .in(data),
27     .out(result3),
28     .v(v)
29 );
30
31 initial begin
32     data=4'b0000;
33 end
34
35 always begin
36     #5;
37     if (data>15) begin
38         data=0;
39     end
40     data=data+1;
41 end
42
43 endmodule

```